HW 12 Mathematics 503, computer part, July 26, 2007

Nasser M. Abbasi

June 15, 2014

1 Derivation for the Ax=b

This is a suplement to the report for the computer project for Math 503. This includes the symbolic derivation of the A matrix and the b vector for the problem of $Ax=b$ which is generated from the FEM formulation for this project. I also include a very short Mathematica program which implements the FEM solution.

For $x=[0,L]$ where L is the length, we define the shape functions (called tent function in this case) as shown below



The shape function is defined by $\phi_i(x)=\psi\left(\frac{x}{h}-i\right)$ where

$$\psi(z)=\begin{cases}1-z & 0<z<1\\ 0 & z>1\end{cases} \tag{1}$$

And $\psi(z)=\psi(-z)$ as shown in this diagram



The shape function is defined by $\phi_i(x) = \psi\left(\frac{x}{h} - i\right)$

Now the derivative of $\phi_i'(x)$ is given by

$$\phi_i'(x) = \begin{cases} \frac{1}{h} & (i-1)h < x \leq i\,h \\ -\frac{1}{h} & i\,h < x < (i+1)\,h \\ 0 & otherwise \end{cases}$$

Now we write the weak form in terms of the above shape function (which is our admissible direction). From part 1 we had

$$I = \int_0^L y'(x)\,\phi'(x) + q\,y(x)\,\phi(x) - f\,\phi(x)\;\;dx = 0$$

And Let

$$y(x) = \sum_{j=1}^{N} c_j \phi_j(x)$$
$$y'(x) = \sum_{j=1}^{N} c_j \phi_j'(x)$$

Hence, now we pick one admissible direction at a time, and need to satisfy the above integral for each of these. Hence we write

$$I_j = \int_0^L \left(\sum_{i=1}^{N} c_i \phi_i'(x)\right) \phi_j'(x) + q\,\left(\sum_{i=1}^{N} c_i \phi_i(x)\right) \phi_j(x) - f\,\phi_j(x)\;\;dx = 0 \qquad j = 1, 2, \cdots N$$

But due to sphere on influence of the $\phi_j(x)$ extending to only $x_{j-1} \cdots x_{j+1}$ the above becomes

$$I_j = \int_{x_{j-1}}^{x_{j+1}} \left(\sum_{i=j-1}^{j+1} c_i \phi_i'(x)\right) \phi_j'(x) + q\,\left(\sum_{i=j-1}^{j+1} c_i \phi_i(x)\right) \phi_j(x) - f\,\phi_j(x)\;\;dx = 0 \qquad j = 1, 2, \cdots N$$

Hence we obtain N equations which we solve for the N coefficients c_j

Now to evaluate I_j we write

$$
\begin{aligned}
I_j &= \int_{x_{j-1}}^{x_j} \cdots\ dx + \int_{xj}^{x_{j+1}} \cdots\ dx \\
&= \int_{x_{j-1}}^{x_j} \left(\sum_{i=j-1}^{j} c_i \phi_i'(x) \right) \phi_j'(x) + q\ \left(\sum_{i=j-1}^{j} c_i \phi_i(x) \right) \phi_j(x) - f\ \phi_j(x)\ dx \\
&+ \\
&\int_{x_j}^{x_j+1} \left(\sum_{i=j}^{j+1} c_i \phi_i'(x) \right) \phi_j'(x) + q\ \left(\sum_{i=j}^{j+1} c_i \phi_i(x) \right) \phi_j(x) - f\ \phi_j(x)\ dx
\end{aligned}
$$

Now we will show the above for $j = 1$ which will be sufficient to build the A matrix due to symmetry.

$\boxed{\text{For } j = 1}$

$$
I_1 = \int_0^{2h} \left(\sum_{i=1}^{2} c_i \phi_i'(x) \right) \phi_1'(x) + q\ \left(\sum_{i=1}^{2} c_i \phi_i(x) \right) \phi_1(x) - f\ \phi_1(x)\ \ dx
$$

Hence breaking the interval into 2 parts we obtain

$$
\begin{aligned}
I_1 &= \int_0^{h} \left(\sum_{i=1}^{1} c_i \phi_i'(x) \right) \phi_1'(x) + q\ \left(\sum_{i=1}^{1} c_i \phi_i(x) \right) \phi_1(x) - f\ \phi_1(x)\ \ dx \\
&+ \\
&\int_h^{2h} \left(\sum_{i=1}^{2} c_i \phi_i'(x) \right) \phi_1'(x) + q\ \left(\sum_{i=1}^{2} c_i \phi_i(x) \right) \phi_1(x) - f\ \phi_1(x)\ dx
\end{aligned}
$$

Hence

$$
\begin{aligned}
I_1 &= \int_0^{h} \left(c_1 \phi_1'(x) \right) \phi_1'(x) + q\ \left(c_1 \phi_1(x) \right) \phi_1(x) - f\ \phi_1(x)\ \ dx \\
&+ \\
&\int_h^{2h} \left(c_1 \phi_1'(x) + c_2 \phi_2'(x) \right) \phi_1'(x) + q\ \left(c_1 \phi_1(x) + c_2 \phi_2(x) \right) \phi_1(x) - f\ \phi_1(x)\ \ dx
\end{aligned} \tag{2}
$$

Now set up a little table to do the above integral.

Range	ϕ_1'	ϕ_2'	ϕ_1	ϕ_2
$[0,h]$	$\frac{1}{h}$	N/A	$\psi\left(-\frac{x}{h}+1\right) \to \frac{x}{h}$	N/A
$[h,2h]$	$\frac{-1}{h}$	$\frac{1}{h}$	$\psi\left(\frac{x}{h}-1\right) \to 2-\frac{x}{h}$	$\psi\left(-\frac{x}{h}+2\right) \to \frac{x}{h}-1$

The above table was build by noting that for ϕ_j, it will have the equation $\psi\left(\frac{x}{h}-i\right)$ when x is under the left leg of tent. And it will have the equation $\psi\left(-\frac{x}{h}+i\right)$ when x is under the right leg of the tent.

This is because for $x<0,$ the argument to $\psi()$ is negative and so we flip the argument as per the definition for ψ shown in the top of this report.

Hence we obtain for the integral in (2)

$$
\begin{aligned}
I_1 = &\int_0^h \left[c_1\left(\frac{1}{h}\right)\right]\left(\frac{1}{h}\right)+q\ \left(c_1\frac{x}{h}\right)\frac{x}{h}-f\ \frac{x}{h}\ dx \\
&+ \\
&\int_h^{2h}\left[c_1\left(\frac{-1}{h}\right)+c_2\ \left(\frac{1}{h}\right)\right]\ \left(\frac{-1}{h}\right)+q\ \left(c_1\left(2-\frac{x}{h}\right)+c_2\left(\frac{x}{h}-1\right)\right)\left(2-\frac{x}{h}\right)-f\left(\ 2-\frac{x}{h}\right)\ dx
\end{aligned}
$$

so the above becomes integral becomes

$$
\begin{aligned}
I_1 = &\int_0^h \frac{c_1}{h^2}+q\ c_1\left(\frac{x^2}{h^2}\right)-f\frac{x}{h}\ dx \\
&+ \\
&\int_h^{2h}\frac{c_1}{h^2}-\ \frac{c_2}{h^2}+qc_1\left(4-4\frac{x}{h}+\frac{x^2}{h^2}\right)+qc_2\left(3\frac{x}{h}-\frac{x^2}{h^2}-2\right)-2f\ +f\frac{x}{h}\ dx
\end{aligned}
$$

Hence

$$
\begin{aligned}
I_1 = &\frac{c_1}{h^2}\int_0^h dx+\frac{q}{h^2}\ c_1\int_0^h x^2dx-\frac{f}{h}\ \int_0^h xdx \\
&+ \\
&\frac{c_1}{h^2}\int_h^{2h}dx-\ \frac{c_2}{h^2}\int_h^{2h}dx+qc_1\int_h^{2h}\left(4-4\frac{x}{h}+\frac{x^2}{h^2}\right)dx+qc_2\int_h^{2h}\left(3\frac{x}{h}-\frac{x^2}{h^2}-2\right)dx-2f\int_h^{2h}dx\ +\frac{f}{h}\int_h^{2h}x\ \ dx
\end{aligned}
$$

Which becomes

$$
\begin{aligned}
I_1 = &\frac{c_1}{h^2}h+\frac{q}{h^2}\ c_1\left(\frac{1}{3}h^3\right)-\frac{f}{2h}h^2 \\
&+ \\
&\frac{c_1}{h^2}h-\ \frac{c_2}{h^2}h+qc_1\left(4h-2\frac{3h^2}{h}+\frac{1}{3}\frac{7h^3}{h^2}\right)+qc_2\left(\frac{3}{2h}\left(3h^2\right)-\frac{1}{3h^2}\left(7h^3\right)-2h\right)-2fh\ +\frac{f}{2h}3h^2
\end{aligned}
$$

or

$$
\begin{aligned}
I_1 = &\frac{c_1}{h}+c_1\ \frac{qh}{3}-\frac{f}{2}h \\
&+ \\
&\frac{c_1}{h}-\ \frac{c_2}{h}+qc_1\left(4h-6h+\frac{7}{3}h\right)+qc_2\left(\frac{9h}{2}-\frac{7}{3}h-2h\right)-2fh\ +\frac{3f}{2}h
\end{aligned}
$$

Therefore

$$I_1 = \frac{c_1}{h} + c_1\,\frac{qh}{3} + \frac{c_1}{h} - \,\frac{c_2}{h} + qc_1\left(\frac{1}{3}h\right) + qc_2\left(\frac{1}{6}h\right) - fh$$

Hence

$$I_1 = c_1\left(\frac{2}{h} + \frac{2}{3}qh\right) + c_2\left(-\frac{1}{h} + \frac{1}{6}qh\right) - fh = 0$$

Multiply by h we obtain

$$\boxed{I_1 = c_1\left(2 + \tfrac{2}{3}h^2 q\right) + c_2\left(-1 + \tfrac{1}{6}h^2 q\right) - fh^2 = 0} \tag{2}$$

Hence we now can set up the $Ax = b$ system using only the above equation by taking advantage that A will be tridiagonal and there is symmetry along the diagonal.

$$\begin{bmatrix} \left(2+\frac{2}{3}h^2q\right) & \left(-1+\frac{1}{6}h^2q\right) & 0 & 0 & \cdots & 0 \\ \left(-1+\frac{1}{6}h^2q\right) & \left(2+\frac{2}{3}h^2q\right) & \left(-1+\frac{1}{6}h^2q\right) & 0 & \cdots & 0 \\ 0 & \left(-1+\frac{1}{6}h^2q\right) & \left(2+\frac{2}{3}h^2q\right) & \left(-1+\frac{1}{6}h^2q\right) & \cdots & 0 \\ 0 & 0 & 0 & 0 & \ddots & \vdots \\ 0 & 0 & 0 & \cdots & \left(-1+\frac{1}{6}h^2q\right) & \left(2+\frac{2}{3}h^2q\right) \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \\ c_3 \\ \vdots \\ c_N \end{bmatrix} = fh^2 \begin{bmatrix} 1 \\ 1 \\ 1 \\ \vdots \\ 1 \end{bmatrix}$$

The following is the FEM program to implement the above, with few plots showing how close it gets to the real solution as N increases.

```
In[1815]:= Remove["Global`*"];
(*by Nasser Abbasi. FEM program for Math 503*)

ψ[x_] := 1 - x /; 0 < x < 1
ψ[x_] := 0 /; x > 1
ψ[x_] := ψ[-x] /; x < 1
ϕ[x_, i_, h_] := ψ[x/h - i];

yApprox[x_, c_, h_, n_] := Module[{}, Sum[c[[i]] ϕ[x, i, h], {i, 1, n}]]

plotIt[msg_, n_] := Module[{A, b, c, nElements, nShapeFunctions, i, j, h, q, f, L},
  q = 4; f = 4; L = 1;  nElements = n; nShapeFunctions = nElements - 1; h = L/nElements;
  exactSol = y[x] /. First@DSolve[{-y''[x] + q y[x] == f, y[0] == 0, y[L] == 0}, y[x], x];
  A = Table[If[i == j, 2 + 2/3 q h^2, If[j == i - 1 || j == i + 1, -1 + 1/6 q h^2, 0]],
    {i, nShapeFunctions}, {j, nShapeFunctions}];
  b = Table[h^2 f , {nShapeFunctions}];
  c = LinearSolve[A, b];
  Plot[{exactSol, yApprox[x, c, h, nShapeFunctions]}, {x, 0, L}, PlotRange → All,
   PlotLabel → msg, ImageSize → 250, PlotStyle → {Black, {Dashed, Red, Thickness → .01}}]
 ]

In[1822]:= p = Table[plotIt["N=" <> ToString[i], i], {i, 2, 7}]

Out[1822]=
```



I also written a small Manipulate program to simulate the above. Here it is

```
In[345]:= Remove["Global`*"];
          (*by Nasser Abbasi. FEM program for Math 503*)

          ψ[x_, L_] := L - x /; 0 ≤ x ≤ L
          ψ[x_, L_] := 0 /; x > L
          ψ[x_, L_] := ψ[-x, L] /; x < L
          ϕ[x_, i_, h_, L_] := ψ[x/h - i, L]

          yApprox[x_, c_, h_, n_, L_] := Module[{i}, Sum[c[[i]] ϕ[x, i, h, L], {i, 1, n}]]

          plotIt[n_] :=
           Module[{A, b, c, nElements, nShapeFunctions, i, j, h, q, f, t1, t2, t3, L, y,
             exactSol, grid, p, p2, nPoints, rmserror, data},
            L = 1; q = 4; f = 4;  nElements = n; nShapeFunctions = nElements - 1; h = L/nElements;
            nPoints = nElements + 1;
            exactSol = y[x] /. First@DSolve[{-y''[x] + q y[x] == f, y[0] == 0, y[L] == 0}, y[x], x];
            (*t1=(2/h + 2 h (1-3 L+3 L^2) q/(3 L^2)) ;t2=(-1/h + 1/6 h (6+1/L^2-6/L) q) ;t3=-f h (-2+1/L);*)
            t1 = 2 + 2/3 q h^2; t2 = -1 + 1/6 q h^2; t3 = h^2 f;
            A = Table[If[i == j, t1, If[j == i - 1 || j == i + 1, t2, 0]], {i, nShapeFunctions},
              {j, nShapeFunctions}];
            b = Table[t3, {nShapeFunctions}];
            c = LinearSolve[A, b];
            grid = Range[0, L, h] // N;
            rmserror = Sum[(exactSol /. x -> grid[[i]] - yApprox[grid[[i]], c, h, nShapeFunctions, L])^2, {i, 1, nPoints}];
            rmserror = Sqrt[rmserror] / nPoints;
            p = Plot[{exactSol, yApprox[x, c, h, nShapeFunctions, L]}, {x, 0, L},
              PlotRange -> All,
              PlotLabel -> "N=" <> ToString[n] <> "\nRMS error=" <> ToString[rmserror],
              PlotStyle -> {Black, {Dashed, Red}}, PlotRange -> {{0, 1}, {0, .5}}, AxesOrigin -> {0, 0}];

            data = Table[{grid[[i]], yApprox[grid[[i]], c, h, nShapeFunctions, L]}, {i, 1, nPoints}];
            p2 = Graphics[{PointSize[Large], Point[data]}];
            Show[{p, p2}]

           ]

In[353]:= demo = Manipulate[plotIt[i], {i, 2, 30, 1}, AutorunSequencing -> {{1, 80}},
            FrameLabel ->
             "Finite Element solution, hardcoded A/b method. Math 503, summer 2007,CSUF
               by Nasser Abbasi"]
```

Out[353]=



Finite Element solution, hardcoded A/b method. Math 503, summer 2007,CSUF by Nasser Abbasi